Universidade Federal da Paraíba
Centro de Ciências e Tecnologia
Departamento de Engenharia Elétrica

Aluna: Mona Lisa Mendonça Fagundes
Matrícula: 9611084-7
Orientador: Dagoberto Lourenço Ribeiro

Projeto de Engenharia Elétrica

"Modelo Generalizado de Carga em Sistemas de Potência em Regime Permanente"

Sumário

Dagoberto Lourenço Ribeiro
Orientador

Mona Lisa M. Fagundes.

Mona Lisa Mendonça Fagundes
Aluna

1.OBJETIVOS

Estudar métodos de otimização não-lineares na estimação de parâmetros de modelos é objetivo geral deste projeto. O objetivo específico é desenvolver e implementar em computador um algoritmo para criação de modelos de carga do sistema de distribuição para análise em regime permanente, utilizando a linguagem de programação FORTRAN 90.

2. INTRODUÇÃO

Em estudos de fluxo de carga e de estabilidade é importante conhecer a variação das potências ativa e reativa com a tensão. Segundo PAI (1979) numa barra típica do sistema de potência , a carga compreende em cerca de 50 a 70% de motores de indução, de 20 a 30% de aquecimento e iluminação e apenas de 5 a 10% de motores síncronos.

Embora seja possível considerar as características *P-V* e *P-Q* de cada uma dessas cargas para efeito de simulação, o tratamento analítico seria muito complicado. Para propósito de análise há principalmente três maneiras de representar a carga:

(i)Modelo da potência constante , segundo o qual ambas as potências especificadas, ativa e reativa, são consideradas constantes. Este é o modelo que predomina em estudos de fluxo de carga.

(ii)Modelo da corrente constante. Deste modo, a corrente de carga é calculada como

$$I = \frac{P - jQ}{V^*} = \left| I \angle (\theta - \Phi) \right|$$

Onde $V = |V| \angle \theta$ e $\Phi = \tan^{-1} Q/P$ é o ângulo do fator de potência.

A amplitude da corrente é mantida constante.

(iii) Modelo da impedância constante. Este é o modelo de carga mais usado em estudos de estabilidade. Mantém-se constante a impedância da carga, que é calculada a partir dos MW e MVAR (supondo-se que sejam conhecidos):

$$Z = \frac{V}{I} = \frac{|V|^2}{P - jQ}$$

3. METODOLOGIA

3.1 Formulação do problema

O interesse neste projeto é por um modelo generalizado. Isto é, em vez de se decidir a prior que a carga deve ser representada por potência constante, corrente constante ou impedância constante, como se faz usualmente, se considerará que ela seja representada dos três modos simultaneamente (em partes por potência, por corrente e por impedância constantes), conforme é mostrado na fig.1.



Figura 1- Modelo generalizado de carga.

Para se identificar os parâmetros do modelo serão empregadas as informações que se dispõem com mais facilidade na operação:

- Curva de tensão: *V(t) versus t,*
- Curva de potência ativa: *P(t) versus t,*
- Curva de potência reativa: *Q(t) versus t.*

onde t é o tempo. A partir dessas, se comporão as curvas P(V) e Q(V). Por mera conveniência a tensão será tomada com referência de fase, **V=V∠0**. Como pela lei de Kirchoff dos nós:

$$S(V) = S_0 + VI^* + \frac{V^2}{Z^*}$$

então,

$$P(V) + jQ(V) = P_0 + jQ_0 + VI\angle -\theta + \frac{V^2}{R - jX}$$

Ou,

$$P(V) + jQ(V) = P_0 + jQ_0 + (I\cos\theta)V - j(I\,\text{sen}\,\theta)V + (V^2)/(R^2 + X^2)(R + jX),$$

Ou ainda

$$P(V) + jQ(V) = P_0 + jQ_0 + (I\cos\theta)V - j(I\,\text{sen}\,\theta)V + [R/(R^2 + X^2]V^2 - j[X/(R^2 + X^2)]V^2 \quad (4)$$

Da equação complexa (4) se obtém as duas equações reais seguintes:

$$P(V) = P_0 + (I\cos\theta)V + [R/(R^2 + X^2)]V \quad (5a)$$

$$Q(V) = Q_0 - (I\,\text{sen}\,\theta)V + [X/(R^2 + X^2)]V \quad (5b)$$

Onde P_0, Q_0, I, θ, R,X são os parâmetros do modelo a serem determinados.

As equações (5) evidenciam que o modelo adotado (fig.1) é quadrático. Seus parâmetros podem ser determinados ajustando-se parábolas

$$P(V) = \gamma_P + \beta_P V + \alpha_P V^2 \qquad (6a)$$

$$Q(V) = \gamma_Q + \beta_Q V + \alpha_Q V^2 \qquad (6b)$$

Às seqüências {V,P} e {V,Q} conhecidas.

Empregando-se o método dos mínimos quadrados determinam-se os seis coeficientes nas eqs.(6), os quais uma vez conhecidos possibilitam a determinação dos parâmetros do modelo, pois:

$$P_0 = \gamma_P \qquad (7a)$$
$$Q_0 = \gamma_Q \qquad (7b)$$

$$I \cos\theta = \beta_P \qquad (8a)$$

$$-I \operatorname{sen}\theta = \beta_Q \qquad (8b)$$

$$R/(R^2 + X^2) = \alpha_P \qquad (9a)$$

$$X/(R^2 + X^2) = \alpha_Q \qquad (9b)$$

Combinando-se as equações 8 se determinam

$$I = \sqrt{\beta_P^2 + \beta_Q^2} \tag{10a}$$

$$\theta = -tg^{-1}\beta_Q / \beta_P \tag{10b}$$

Do mesmo modo, combinando-se as eqs. (9) se encontram:

$$R = \alpha_P / (\alpha_P^2 + \alpha_Q^2) \tag{11a}$$

$$X = \alpha_Q / (\alpha_P^2 + \alpha_Q^2) \tag{11b}$$

3.2 Método dos Mínimos Quadrados

Qualquer curva $y=f(x)$ pode se ajustar a um conjunto de pontos $(x_1,y_1), (x_2,y_2),\ldots, (x_m,y_m)$. O critério usual de ajuste é o dos mínimos quadrados, segundo o qual uma curva se ajusta a um conjunto de pontos dados quando a soma dos quadrados dos erros em cada ponto é mínimo. Isto é, se

$$\sum_{i=1}^{m} [y_i - f(x_i)]^2 = E^2$$

e E^2 é mínimo, então a curva $f(x)$ versus x se ajusta à seqüência

$$S = \{(x_1,y_1), (x_2,y_2),\ldots, (x_m,y_m)\}.$$

E^2 é um indicador de qualidade. Quanto menor seu valor, mais fino é o ajuste, de modo que isso pode ser usado para se escolher entre duas curvas qual a que melhor se ajusta a uma sequência.

Como um dos nossos problemas consiste em solucionar sistemas de equações lineares, conforme será visto em 3.3, aplicamos tal método para os sistemas de equações sugeridos na seção 3.1. Eis aqui um breve procedimento para tal aplicação:

Dado um sistema linear da forma $Ax = b$, o procedimento para encontrar a aproximação por mínimos quadrados para o sistema acima é o seguinte:

Etapa 1: Forme os seguintes sistemas: A^TA e A^Tb

Etapa 2: Resolver o sistema normal $A^TAx = A^Tb$ para x usando o método de *redução de Gauss*, onde A^T é a matriz transposta de A.

3.3 Otimização como um problema de solução de sistemas de equações lineares

Diante da necessidade de ajustarmos o nosso conjunto de dados a uma função é preciso estabelecermos o tipo de curva e calcularmos os parâmetros dessa curva. Se conhecermos os parâmetros de uma curva podemos calcular $f(x_i)$ e compararmos com o valor dado y_i.

Neste projeto foi estudado de modo especial, o ajuste polinomial de curvas utilizando o método dos mínimos quadrados, como critério de qualidade do ajuste dos parâmetros. Tal abordagem foi definida, a partir da formulação do problema, bem com da definição dos parâmetros calculados para o modelo da carga. É visto que na seção 4.1, o problema se resume à um ajuste de curva polinomial aplicado na resolução de sistemas de equações lineares.

3.3.1 Ajuste polinomial

Suponha-se que se tenha n pontos representando dados $(x_1,y_1), (x_2,y_2),\ldots, (x_m,y_m)$ onde pelo menos $m+1$ dos x_i sejam distintos, o nosso objetivo é construir um modelo matemático da forma:

$$u(x) = a_0 + a_1x + a_2x^2 + \ldots + a_mx^n \ , \qquad m < n$$

que 'melhor se ajuste' a esses dados.

O problema para determinar os coeficientes do polinômio(acima) de ordem n que se ajuste aos m pontos dados, (x_i, y_i), $i=1, \dots, m$, resume-se em resolver o seguinte sistema de equações lineares:

$$\begin{array}{l}
\mathbf{C_{00}a_0 + C_{01}a_1 + \dots + C_{0n}a_n = B_o} \\
\mathbf{C_{10}a_0 + C_{11}a_1 + \dots + C_{1n}a_n = B_1} \\
\vdots \qquad \vdots \qquad \vdots \qquad \vdots \\
\mathbf{C_{no}a_0 + C_{n1}a_1 + \dots + C_{nn}a_n = B_n}
\end{array}$$

Onde,

$$C_{ij} = \sum_{k=1}^{m} x_k^{\,i+j} \qquad\qquad B_i = \sum_{k=1}^{m} x_k^{\,i} y_k$$

Nas equacões acima, $i,j = 0,1, \dots ,n$ e $m>n$

3.4 Solução de sistemas de equações lineares:

Diante da necessidade de resolver sistemas de equações lineares encontrados no projeto, foi feito um estudo dos métodos de resolução de sistemas de equações lineares, bem como suas aplicações.

Há diferentes métodos numéricos de resolver um sistema de equações lineares. Um deles, o *método de eliminação de Gauss*, que é bastante útil e de eficiência comprovada. Baseado nisto, tal método foi aplicado ao projeto.

Dado um sistema linear o objetivo é manipular a matriz aumentada representando o sistema linear dado até chegar-se à uma forma simplificada a partir da qual a solução passa a ser facilmente encontrada.

A base da *eliminação gaussiana* está no seguinte teorema:

TEOREMA: Seja $Ax = b$ e $Cx = d$, dois sistemas lineares com m equações e n incógnitas. Se as matrizes aumentadas $|A \vdots b|$ e $|C \vdots d|$ desses sistemas são equivalentes por linha, então os dois sistemas tem exatamente as mesmas soluções.(MARINS, 1989)

O algoritmo de redução de *Gauss* que foi utilizado para resolver os sistemas lineares da forma Ax=b do projeto é o seguinte:

Etapa1: Forme a matriz aumentada $|A \vdots b|$;

Etapa 2: Transforme a matriz aumentada à sua forma escada reduzida por linhas usando operações elementares em suas linhas;

Etapa 3: A linha que estiver sendo processada será denominada *linha pivô*. O elemento diagonal da linha pivô será denominado de *elemento pivô*. Inicialmente, se faz a primeira linha com linha pivô, de modo que o elemento pivô é a_{11}.

Etapa 4: Fazer a *normalização* do elemento pivô;

Etapa 5: Processo de *redução de linhas*, que tem como objetivo anular os elementos abaixo (e se necessário acima) do elemento pivô. Esse objetivo é alcançado fazendo as operações elementares com as outras linhas subsequentes à linha pivô.

Etapa 6: Processo de *pivotamento por linha ou por coluna* a fim de contornar o problema de o elemento pivô ser zero, ou muito pequeno em valor absoluto o que implicaria no erro "divisão por zero" (*overflow*).

Etapa 7: Identificação do tipo de sistema (determinado, indeterminado ou inconsistente), a fim de advertir o usuário sobre seu resposta final.

3.5 Algoritmo do método aplicado

Neste projeto, conforme previsto em seu objetivo foi desenvolvido uma rotina computacional capaz de determinar os parâmetros do modelo generalizado de carga, baseado em técnicas de otimização aplicadas na resolução de sistemas de equações lineares.

Uma rotina computacional foi desenvolvida em FORTRAN 90 baseada no seguinte algoritmo:

1. Identifique as variáveis que serão utilizadas no programa;
2. Abra o arquivo de dados, leia os dados e dimensione as variáveis definidas anteriormente;
3. Forme a matriz dos coeficientes do sistema de equações lineares;
4. Forme a matriz (a), vetores independentes (b_p e b_q) e matriz aumentada (aum);
5. Solução do sistema de equações lineares para determinação dos parâmetros do modelo de carga;
6. Início da SUBROUTINE GAUSS para resolução do sistema de equações lineares previstos no projeto;

   6.1 Declaração das variáveis utilizadas na subroutina Gauss;

   6.2 Processo de triangularização da matriz aumentada;

   6.3 Procura da linha mais adequada para ser a *linha pivô*, para se evitar o problema de overflow (ver seção 3.4);

   6.4 Pivotamento de linhas;

   6.5 *Normalização* da linha pivô;

   6.6 Operações elementares com as linhas, a fim de se fazer à redução de linhas;

   6.7 Advertência quanto à natureza do sistema final calculado (ver seção 3.4);

7. Determinação dos parâmetros do modelo;
8. Fim da SUBROUTINA GAUSS.

4 ROTINA COMPUTACIONAL DESENVOLVIDA NO PROJETO

```
! Solução de Sistemas de Equações Lineares pelo Método de Redução de Gauss.

!Debug
program SistemasLineares

!Especificações.
implicit none
real:: meuZero
real::pequeno                !Variável equivalente ao 0(zero).
real::alfap                  !Coeficiente independente de V(tensão)referente a p.
real::betap                  !Coeficiente linear de V(tensão) referente a p.
real::gamap                  !Coeficiente quadrático de V(tensão) referente a p.
real::alfaq                  !Coeficiente independente de V(tensão) referente a q.
real::betaq                  !Coeficiente linear de V(tensão) referente a q.
real::gamaq                  !Coeficiente quadrático de V(tensão) referente a q.
real::P0                     !Potência ativa
real::Q0                     !Potência reativa
real::It                     !Corrente
real::teta                   !Ângulo
real::R                      !Resistência
real::X                      !Reatância

integer,parameter:: n=2      !Ordem do polinômio
real,allocatable:: c(:,:)    !Matriz dos coeficientes
real,allocatable:: a(:,:)    !Matriz da multiplicação entre Ct(transposta de c) e c.
real,allocatable:: aum(:,:)  !Matriz aumentada.
```

```
real,allocatable:: bp(:)         !Coeficiente referentes a potência ativa(p).
real,allocatable:: bq(:)         !Coeficiente referentes a potência ativa(q).
real,allocatable:: v(:)          !Tensões medidas.
real,allocatable:: p(:)          !Potências ativas medidas.
real,allocatable:: q(:)          !Potências reativas medidas.
real,allocatable:: paramp(:)     !Equivale a bp.
real,allocatable:: paramq(:)     !Equivale a bq.
integer:: i                      !Variável inteira.
integer:: m                      !Número de pontos medidos

!Abertura de arquivo
!Dimensionamentos da matriz aumentada e da matriz do termo independente.
open(unit=1,file='sistem2.dat',status='old')
open(unit=2,file='sistem2.sai')

!Leitura do número de pontos medidos(m)
read(1,'(/i3///)') m

!Dimensionamento das matrizes
allocate(c(m,0:n),a(n+1,0:n),aum(n+1,0:n+1),bp(n+1),bq(n+1),v(m),p(m),q(m),
         paramp(n+1),paramq(n+1))

!Definição de parâmetros.
meuZero=10*epsilon(a(1,1))
pequeno=tiny(a(1,1))

!Leitura dos pontos medidos.
read(1,'(f7.3,6x,f5.3,7x,f5.3)')(v(i),p(i),q(i), i=1,m)

!Formação da matriz 'C'.
 C(:,0)=1
      do i=1,n
```

```
        C(:,i)=C(:,i-1)*v
    end do

    !Formação da matriz(a),vetor independente(bp) e matriz aumentada(aum).
        a=matmul(transpose(c),c)
        bp=matmul(transpose(c),p)
        aum(:,0:n)=a
        aum(:,n+1)=bp

    !Solução do sistema de equações lineares para determinações dos parâmetros(alfa(p),
    !beta(p),gama(p)).
    call gauss(aum)
    paramp=aum(:,n+1)
    write(2,'(a)')'Coeficientes referentes a potência ativa(p),(alfa,beta,gama)'
    write(2,*)paramp
    alfap=aum(1,n+1)
    betap=aum(2,n+1)
    gamap=aum(3,n+1)

    !Formação do vetor independente bq e da matriz aumentada(aum).
    bq=matmul(transpose(c),q)
    aum(:,0:n)=a
    aum(:,n+1)=bq

    !Solução do sistema de equações lineares para determinações dos parâmetros(alfa(q),
    !beta(q),gama(q)).
    call gauss(aum)
    paramq=aum(:,n+1)
    write(2,'(a)')'Coeficientes referentes a potência reativa(q),(alfa,beta,gama)'
    write(2,*)paramq
    alfaq=aum(1,n+1)
    betaq=aum(2,n+1)
```

```
gamaq=aum(3,n+1)

!Determinação dos parâmetros do modelo
write(2,*)
write(2,*)'Os Parâmetros do modelo generalizado de carga são:'
P0=alfap
write(2,*)
write(2,*)'O valor de P0 e:'
write(2,*)P0
Q0=alfaq
write(2,*)
write(2,*)'O valor de Q0 e:'
write(2,*)Q0
It=sqrt(betap**2+betaq**2)
write(2,*)
write(2,*)'O valor de It e:'
write(2,*)It
teta=-atan(betaq/betap)
write(2,*)
write(2,*)'O valor de teta e:'
write(2,*)teta
R=alfap/(alfap**2+alfaq**2)
write(2,*)
write(2,*)'O valor de R e:'
write(2,*)R
X=alfaq/(alfap**2+alfaq**2)
write(2,*)
write(2,*)'O valor de X e:'
write(2,*)X
```

contains

```
  subroutine gauss(a)
    real  :: a(n,n+1)
    real  :: linhaCopia(n+1)        !uma linha da matriz aumentada
    real  :: parteLinhaCopia(n)     !uma linha da matriz dos coeficientes.
    real  :: elemPivo,elemAlvo
    logical:: auxiliar
    integer:: linPivo,linAlvo,linMaior


!triangularização da matriz aumentada: pivotamento, normalização e
!redução dos elementos abaixo da diagonal.
 progressiva: do linPivo=1,n

!procura da linha mais adequada para ser pivo.
      elemPivo=0.
      do i=linPivo,n
          if(abs(a(i,linPivo))>elemPivo) then
                elemPivo=a(i,linPivo)
                 linMaior=i
          endif
      end do
      if(abs(elemPivo)<=pequeno) then
        write(2,'(a)') 'O processo é impossível de continuar com pivotamento de linha
apenas!'
        stop 'O processo é impossível de continuar com pivotamento de linha apenas!'
      endif

 !pivotamento de linha.
      linhaCopia=a(linPivo,:)
      a(linPivo,:)=a(linMaior,:)
```

```
      a(linMaior,:)=linhaCopia

!normalização da linha pivo.
      a(linPivo,:)=a(linPivo,:)/elemPivo

!redução das linhas alvos inferiores.
      do linAlvo=linPivo+1,n
        elemAlvo=a(linAlvo,linPivo)
        a(linAlvo,:)=a(linAlvo,:)-a(linPivo,:)*elemAlvo
        parteLinhaCopia=abs(a(linAlvo,1:n))
        auxiliar=all(parteLinhaCopia<meuZero)
        if(auxiliar)then
          if(abs(a(linAlvo,n+1))<meuZero)then
                                    write(2,'(a)')' Sistema indeterminado!'
            write(*,'(a)')' Sistema indeterminado!'
            return
          else
            write(2,'(a)')' Sistema inconsistente!'
            write(*,'(a)')' Sistema inconsistente!'
            return
          end if
        end if
      end do
    enddo progressiva

!redução das linhas alvos superiores.
    regressiva: do linPivo=2,n
      do linAlvo=1,linPivo-1
        elemAlvo=a(linAlvo,linPivo)
        a(linAlvo,:)=a(linAlvo,:)-a(linPivo,:)*elemAlvo
      end do
    enddo regressiva
```

```
  end subroutine gauss
end program sistemasLineares
```

4.1 Arquivo de dados (sistem2.dat):

Número de pontos medidos

48

===

v(KV)	p(MW)	q(Mvar)
13.855	4.605	1.807
13.800	4.545	1.853
13.814	4.559	1.877
13.772	4.427	1.866
13.869	4.655	1.630
13.841	4.705	1.430
13.814	4.600	1.740
13.745	4.368	1.809
13.800	4.682	1.820
13.800	4.778	1.839
13.828	4.596	1.960
13.841	4.495	1.869
13.855	4.336	1.915
13.745	4.070	1.954
13.924	3.110	1.280
13.896	2.822	1.188
13.938	3.049	0.960
13.952	2.860	1.219
14.008	2.816	1.183
13.910	2.857	1.160
13.980	2.915	1.236

13.868	2.854	1.246
13.966	3.540	1.040
14.008	3.020	1.230
13.994	2.877	1.218
13.966	3.058	1.258
13.994	3.043	1.200
13.896	2.877	1.400
13.966	2.816	1.171
13.966	2.630	1.181
13.910	2.836	1.223
14.008	3.046	1.250
13.924	3.020	1.219
13.952	2.996	1.229
13.966	2.816	1.240
13.924	3.220	1.161
13.689	6.657	2.774
13.717	6.716	2.960
13.662	6.700	2.793
13.730	6.539	2.692
13.703	6.370	2.717
13.828	4.700	1.817
13.772	4.500	1.813
13.772	4.368	1.981
13.786	4.323	1.973
13.828	4.418	2.130
13.828	4.545	1.907
13.814	4.359	1.886

4.2 Arquivo de saída (sistem2.sai) ('Resultados obtidos no projeto'):

Coeficientes referentes a potência ativa(p),(alfa,beta,gama)

192.694000 2666.699000 36906.310000

Coeficientes referentes a potência reativa(q),(alfa,beta,gama)

79.304010 1097.344000 15184.890000

Os Parâmetros do modelo generalizado de carga são:

O valor de P0 e:

192.694000

O valor de Q0 e:

79.304010

O valor de It e:

2883.652000

O valor de teta e:

-3.903798E-01

O valor de R e:

4.437899E-03

O valor de X e:

1.826436E-03

5 CONCLUSÃO

Uma rotina computacional capaz de criar o modelo de uma carga a partir dos dados mais disponíveis na operação do sistema foi desenvolvida e implementada. Técnicas de otimização foram empregadas para identificação dos parâmetros do modelo.

A rotina foi aplicada a casos reais . Foram utilizados dados do sistema CEAL (Com-
a ser analisado.
panhia de Eletricidade de Alagoas) e SAELPA(Sociedade de Eletrificação da Paraíba S/A).

Um estudo teórico preliminar de alguns dos principais métodos de otimização foi realizado. Nesse estudo foram contemplados os princípios de cada método, condições e velocidade de convergência, além de precisão e dependência da estimativa inicial.

Os resultados obtidos para os parâmetros do modelo generalizado de carga foram satisfatórios.

6. BIBLIOGRAFIA

ETO, J., "Metered residential cooling loads: models comparision", *IEEE Trans. On Power Systems*, May 1997, pp. 858-868.

HEYDT, G.T., "Distribuition network time-varying loads, periodic steady-state calculation", *IEEE Trans. On Power Systems,* Oct. 1996, pp. 1860-1867.

KHODABAKHCHIAN, B., "Mixed residential-commercial load modeling: disturbe simulation", *IEEE Trans. On Power System,* May 1997, pp. 791-796.

PAI, M. A., "Computer Techniques in Power Systems Analysis". New Delhi: Tata McGrawHill Publishing Co. Ltd., 1979.

TALESKI, R., "Distribuition network reconfiguration: energy loss reduction applications", *IEEE Trans. On Power Systems,* May 1997, pp. 1002-1007.

Anton, H. & Rorres, C. "Elementary Linear Algebra: applications version", 7ª. ed., John Willey & Sons. Inc., New York, 1994.

Kolman, B. "Introdução à Àlgebra Linear Com Aplicações", 6ª. ed., Prentice-Hall do Brasil, Rio de Janeiro, 1998.

Brian Hahn, "FORTRAN 90 for Scientists and Engineers", 1994, 368 pages, ISBN 0 340 60034 9.

SOUZA, B. A., "Apostila de Distribuição de Energia Elétrica".

MARINS, J. M., "Cálculo numérico computacional- Teoria e Prática.", 1ª.ed., Ed. Atlas S.A, São Paulo, 1989.